

How can I use the function in excel to perform the desired task?

Authored by
stats writer

July 1, 2024

RECOMMENDED CITATION

stats writer (2024). *How can I use the function in excel to perform the desired task?*. PSYCHOLOGICAL SCALES. Retrieved from <https://scales.arabpsychology.com/?p=163979>

The function in Excel is a useful tool that allows users to perform various tasks efficiently and accurately. By using the function, users can input specific formulas or commands to manipulate data, perform calculations, and generate desired results. This function can be accessed through the formula bar or by using the shortcut keys. It is a powerful tool that can save time and effort in performing repetitive tasks or complex calculations. By understanding how to use the function effectively, users can enhance their productivity and achieve their desired results in Excel.

The **TEXT** function lets you change the way a number appears by applying formatting to it with **format codes**. It's useful in situations where you want to display numbers in a more readable format, or you want to combine numbers with text or symbols.

Note: The TEXT function converts numbers to text, which may make it difficult to reference in later calculations. It's best to keep your original value in one cell, and then use the TEXT function in another cell. Then, if you need to build other formulas, always reference the original value and not the TEXT function result.

Syntax

TEXT(value, format_text)

The **TEXT** function syntax has the following arguments:

Argument Name	Description
value	A numeric value that you want to be converted into text.
format_text	A text string that defines the formatting that you want to be applied to the supplied value.

Overview

In its simplest form, the TEXT function says:

=TEXT(Value you want to format, "Format code you want to apply")

Here are some popular examples, which you can copy directly into Excel to experiment with on your own. Notice the format codes within quotation marks.

Formula	Description
=TEXT(1234.567,"\$#,##0.00")	Currency with a thousands separator and 2 decimals, like \$1,234.57. Note that Excel rounds the value to 2 decimal places.

Formula	Description
=TEXT(TODAY(),"MM/DD/YY")	Today's date in MM/DD/YY format, like 03/14/12
=TEXT(TODAY(),"DDDD")	Today's day of the week, like Monday
=TEXT(NOW(),"H:MM AM/PM")	Current time, like 1:29 PM
=TEXT(0.285,"0.0%")	Percentage, like 28.5%
=TEXT(4.34,"# ?/?")	Fraction, like 4 1/3
=TRIM(TEXT(0.34,"# ?/?"))	Fraction, like 1/3. Note this uses the TRIM function to remove the leading space with a decimal value.
=TEXT(12200000,"0.00E+00")	Scientific notation, like 1.22E+07
=TEXT(1234567898,"###-####;(###) ###-####")	Special (Phone number), like (123) 456-7898
=TEXT(1234,"0000000")	Add leading zeros (0), like 0001234
=TEXT(123456,"##0° 00' 00'")	Custom - Latitude/Longitude

Note: Although you can use the TEXT function to change formatting, it's not the only way. You can



change the format without a formula by pressing **CTRL+1** (or **+1** on the Mac), then pick the format you want from the **Format Cells > Number** dialog box.

Download our examples

You can download an example workbook with all of the TEXT function examples you'll find in this article, plus some extras. You can follow along, or create your own TEXT function format codes.

[Download Excel TEXT function examples](#)

Format codes by category


Following are some examples of how you can apply different number formats to your values by using the **Format Cells** dialog box, and then use the **Custom** option to copy those **format codes** to your **TEXT** function.

Select a number format? Select a number format
 Leading zero's (0's)
 Display a thousands separator
 Number, currency and accounting formats
 Dates
 Times
 Percentages
 Fractions
 Scientific notation
 Special formats

Why does Excel delete my leading 0's?

Excel is trained to look for numbers being entered in cells, not numbers that look like text, like part numbers or SKU's. To retain leading zeros, format the input range as Text before you paste or enter values. Select the column, or range where you'll be putting the values, then use **CTRL+1** to bring up the **Format > Cells** dialog and on the **Number tab** select **Text**. Now Excel will keep your leading 0's.

If you've already entered data and Excel has removed your leading 0's, you can use the **TEXT** function to add them back. You can reference the top cell with the values and use **=TEXT(value,"00000")**, where the number of 0's in the formula represents the total number of characters you want, then copy and paste to the rest of your range.



B	C	D	E
Original Value	0's removed	TEXT function	Convert back to Numbers
00001	1	1 00001	1
00012	12	12 00012	12
00123	123	123 00123	123
01234	1234	1234 01234	1234
12345	12345	12345 12345	12345

If for some reason you need to convert text values back to numbers you can multiply by 1, like **=D4*1**, or use the double-unary operator (--), like **==D4**.

Excel separates thousands by commas if the format contains a comma (,) that is enclosed by number signs (#) or by zeros. For example, if the format string is **"#,###"**, Excel displays the number 12200000 as 12,200,000.

A comma that follows a digit placeholder scales the number by 1,000. For example, if the format string is **"#,###.0"**, Excel displays the number 12200000 as 12,200.0.

	A	B	C
3	Value	Formula	Result
4	12200000	=TEXT(A4,"#,###")	12,200,000
5	12200000	=TEXT(A5,"0,000.00")	12,200,000.00
6	12200000	=TEXT(A6,"#,")	12200
7	12200000	=TEXT(A7,"#,###.0,")	12,200.0
8	12200000	=TEXT(A8,"0.0,,")	12.2

Notes:

The thousands separator is dependent on your regional settings. In the US it's a comma, but in other locales it might be a period (.).


The thousands separator is available for the number, currency and accounting formats.

Following are examples of standard number (thousands separator and decimals only), currency and accounting formats. Currency format allows you to insert the currency symbol of your choice and aligns it next to your value, while accounting format will align the currency symbol to the left of the cell and the value to the right. Note the difference between the currency and accounting format codes below, where accounting uses an asterisk (*) to create separation between the symbol and the value.


	A	C	D
1	Value	Formula	Result
2	1234.56	=TEXT(A2,"0.00")	1234.56
3	1234.56	=TEXT(A3,"#,##0")	1,235
4	-1234.56	=TEXT(A4,"#,##0.00")	-1,234.56
5	1234.56	=TEXT(A5,"\$#,##0")	\$1,235
6	1234.56	=TEXT(A6,"\$#,##0.00")	\$1,234.56
7	-1234.56	=TEXT(A7,"\$#,##0.00_);(\$#,##0.00)")	(\$1,234.56)
8	1234.56	=TEXT(A8,"\$ * #,##0")	\$ 1,235
9	1234.56	=TEXT(A9,"\$ * #,##0.00")	\$ 1,234.56
10	-1234.56	=TEXT(A10,"\$ * #,##0.00_);(\$ * #,##0.00)")	(\$ 1,234.56)



To find the format code for a currency symbol, first press **Ctrl+1** (or **+1** on the Mac), select the format you want, then choose a symbol from the **Symbol** drop-down:



Then click **Custom** on the left from the **Category** section, and copy the format code, including the currency symbol.




Note: The TEXT function does not support color formatting, so if you copy a number format code from the Format Cells dialog that includes a color, like this: \$#,##0.00_);(\$#,##0.00), the TEXT function will accept the format code, but it won't display the color.

You can alter the way a date displays by using a mix of "M" for month, "D" for days, and "Y" for years.

	A	B	C	D	E
2					
3	Date:	10/12/2016			
4					
5	To display	As	Format	Formula	Result
6	Months	1-12	"m"	=TEXT(B3,"m")	10
7	Months	01-12	"mm"	=TEXT(B3,"mm")	10
8	Months	Jan-Dec	"mmm"	=TEXT(B3,"mmm")	Oct
9	Months	January-December	"mmmm"	=TEXT(B3,"mmmm")	October
10	Months	J-D	"mmmmm"	=TEXT(B3,"mmmmm")	O
11	Days	1-31	"d"	=TEXT(B3,"d")	12
12	Days	01-31	"dd"	=TEXT(B3,"dd")	12
13	Days	Sun-Sat	"ddd"	=TEXT(B3,"ddd")	Wed
14	Days	Sunday-Saturday	"dddd"	=TEXT(B3,"dddd")	Wednesday
15	Years	00-99	"yy"	=TEXT(B3,"yy")	16
16	Years	1900-9999	"yyyy"	=TEXT(B3,"yyyy")	2016

Format codes in the TEXT function aren't case sensitive, so you can use either "M" or "m", "D" or "d", "Y" or "y".

	<p>Mynda Recommends...</p>	<p>If you share Excel files and reports with users from different countries, then you might want to give them a report in their language. Excel MVP, Mynda Treacy has a great solution in this Excel Dates Displayed in Different Languages article. It also includes a sample workbook you can download.</p>
---	-----------------------------------	---

You can alter the way time displays by using a mix of "H" for hours, "M" for minutes, or "S" for seconds, and "AM/PM" for a 12-hour clock.

	A	B	C	D	E
2					
3	Current time:	2:22 PM			
4					
5	To display	As	Format	Formula	Result
6	Hours	0-23	"h"	=TEXT(B3,"h")	14
7	Hours	00-23	"hh"	=TEXT(B3,"hh")	14
8	Minutes	0-59	"m"	=TEXT(B3,"m")	10
9	Minutes	00-59	"mm"	=TEXT(B3,"mm")	10
10	Seconds	0-59	"s"	=TEXT(B3,"s")	44
11	Seconds	00-59	"ss"	=TEXT(B3,"ss")	44
12	Time	2 PM	"h AM/PM"	=TEXT(B3,"h AM/PM")	2 PM
13	Time	2:22 PM	"h:mm AM/PM"	=TEXT(B3,"h:mm AM/PM")	2:22 PM
14	Time	2:22:44 P	"h:mm:ss A/P"	=TEXT(B3,"h:mm:ss A/P")	2:22:44 P
15	Time	14:22:44.11	"h:mm:ss.00"	=TEXT(B3,"h:mm:ss.00")	14:22:44.11
16	Elapsed Time (hours & minutes)	1:02	"[h]:mm"	=TEXT(B16,"[h]:mm")	1:02
17	Elapsed Time (minutes & seconds)	62:16	"[mm]:ss"	=TEXT(B17,"[mm]:ss")	62:16
18	Elapsed Time (seconds & hundredths)	3735.80	"[ss].00"	=TEXT(B18,"[ss].00")	3735.80

If you leave out the "AM/PM" or "A/P", then time will display based on a 24-hour clock.

Format codes in the TEXT function aren't case sensitive, so you can use either "H" or "h", "M" or "m", "S" or "s", "AM/PM" or "am/pm".

You can alter the way decimal values display with percentage (%) formats.

	A	B	C	D
3	Value	Description	Formula	Result
4	0.244740088	No decimals	=TEXT(A4,"0%")	24%
5	0.244740088	1 decimal	=TEXT(A5,"0.0%")	24.5%
6	0.244740088	2 decimals	=TEXT(A6,"0.00%")	24.47%

You can alter the way decimal values display with fraction (/?/?) formats.

	A	B	C	D
3	Value	Description	Formula	Result
4	4.340381685	Up to one digit (1/4)	=TEXT(A4,"# ?/?")	4 1/3
5	4.340381685	Up to two digits (21/25)	=TEXT(A5,"# ??/??")	4 16/47
6	4.340381685	Up to three digits (312/943)	=TEXT(A6,"# ???/???)	4 161/473
7	4.340381685	As halves (1/2)	=TEXT(A7,"# ?/2")	4 1/2
8	4.340381685	As quarters (2/4)	=TEXT(A8,"# ?/4")	4 1/4
9	4.340381685	As sixteenths (8/16)	=TEXT(A9,"# ??/16")	4 5/16
10	4.340381685	As tenths (3/10)	=TEXT(A10,"# ?/10")	4 3/10
11	4.340381685	As hundreths (30/100)	=TEXT(A11,"# ??/100")	4 34/100

Scientific notation is a way of displaying numbers in terms of a decimal between 1 and 10, multiplied by a power of 10. It is often used to shorten the way that large numbers display.

	A	B	C	D
3	Value	Description	Formula	Result
4	12,200,000	Scientific - 7 places	=TEXT(A4,"0.00E+00")	1.22E+07
5	12,200,000	Scientific - 6 places	=TEXT(A5,"#0.0E+0")	12.2E+6

Excel provides 4 special formats:

Zip Code - "00000"

Zip Code + 4 - "00000-0000"

Phone Number - "###-####;(###) ###-####"

Social Security Number - "000-00-0000"

	A	B	C	D
3	Value	Description	Formula	Result
4	12345	Zip Code	=TEXT(A4,"00000")	12345
5	123456789	Zip Code + 4	=TEXT(A5,"00000-0000")	12345-6789
6	1234567899	Phone Number	=TEXT(A6,"[<=9999999]###-####;(###) ###-####")	(123) 456-7899
7	123456789	Social Security Number	=TEXT(A7,"000-00-0000")	123-45-6789

Special formats will be different depending on locale, but if there aren't any special formats for your locale, or if these don't meet your needs then you can create your own through the **Format Cells > Custom** dialog.

Common scenario

The **TEXT** function is rarely used by itself, and is most often used in conjunction with something else. Let's say you want to combine text and a number value, like "Report Printed on: 03/14/12", or "Weekly Revenue: \$66,348.72". You could type that into Excel manually, but that defeats the purpose of having Excel do it for you. Unfortunately, when you combine text and formatted numbers, like dates, times, currency, etc., Excel doesn't know how you want to display them, so it drops the number formatting. This is where the **TEXT** function is invaluable, because it allows you to force Excel to format the values the way you want by using a **format code**, like "**MM/DD/YY**" for date format.

In the following example, you'll see what happens if you try to join text and a number without using the **TEXT** function. In this case, we're using the **ampersand (&)** to concatenate a text string, a space (" "), and a value with **=A2&" "&B2**.

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C
1	Text	Value	Result
2	Report Printed on:	03/14/12	Report Printed on: 40982

The formula bar for cell C2 shows: `=A2&" "&B2`

As you can see, Excel removed the formatting from the date in cell B2. In the next example, you'll see how the **TEXT** function lets you apply the format you want.

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C
1	Text	Value	Result
2	Report Printed on:	03/14/12	Report Printed on: 03/14/12

The formula bar for cell C2 shows: `=A2&" "&TEXT(B2,"mm/dd/yy")`

Our updated formula is:

Cell C2: `=A2&" "&TEXT(B2,"mm/dd/yy")` - Date format

Frequently Asked Questions

Unfortunately, you can't do that with the **TEXT** function; you need to use Visual Basic for Applications (VBA) code. The following link has a method: [How to convert a numeric value into](#)


English words in Excel.

Yes, you can use the [UPPER](#), [LOWER](#) and [PROPER](#) functions. For example, =UPPER("hello") would return "HELLO".

Yes, but it takes a few steps. First, select the cell or cells where you want this to happen and use **Ctrl+1** to bring up the **Format > Cells** dialog box, then **Alignment > Text control** > check the **Wrap Text** option. Next, adjust your completed **TEXT** function to include the ASCII function **CHAR(10)** where you want the line break. You might need to adjust your column width depending on how the final result aligns.

="Today is: "&CHAR(10)&TEXT(TODAY(),"mm/dd/yy")			
C	D	E	F
TEXT w/a Line Break			
Today is: 05/20/16			

In this case, we used: ="Today is: "&CHAR(10)&TEXT(TODAY(),"mm/dd/yy")

	<p>Mynda Recommends...</p> <p>If you share Excel files and reports with users from different countries, then you might want to give them a report in their language. Excel MVP, Mynda Treacy has a great solution in this Excel Dates Displayed in Different Languages article. It also includes a sample workbook you can download.</p>
---	---

See Also

[Create or delete a custom number format](#)

[Convert numbers stored as text to numbers](#)

[All Excel functions \(by category\)](#)